

Applies To:

SAP NetWeaver 2004s – Web Dynpro for ABAP

Summary

This tutorial provides a step-by-step guide for using Select Option functionality in a WDA application. This tutorial assumes that you have completed [WDA Tutorial I: Getting Started with Web Dynpro for ABAP](#) and have a good understanding of the basics of Web Dynpro for ABAP.

By: Rich Heilman

Company: Yorktowne Cabinetry


Date: 15 January 2006

Table of Contents

Applies To:.....	1
Summary	1
Table of Contents	1
Step 1 – Creating the Web Dynpro for ABAP(WDA) Object	2
Step 2 – Define Used Components.....	4
Step 3 – Creating a View.....	5
Step 4 – Defining Attributes.....	9
Step 5 – Method Implementation	10
Step 6 – Embed View into Window	13
Step 7 – Create the Application and Test	14
Author Bio.....	15
Disclaimer & Liability Notice	15

Step 1 – Creating the Web Dynpro for ABAP (WDA) Object

Go to transaction code SE80. This is the ABAP Workbench. In the object list box, choose *Web Dynpro Comp / Intf.* Enter the name of the object as Z_TUTORIAL2 and hit Enter. The system will ask you if you want to create this object. Click Yes.



WDA Tutorial II: Using Select Options in a WDA Application


Enter a description for the object and hit Enter. A dialog will appear asking to assign a package. Click *Local Object*.

The screenshot shows a dialog box titled "Web Dynpro: Component / Create Interface". It contains the following fields and options:

- Name:** Z_TUTORIAL2
- Description:** Tutorial 2
- Type:** Radio buttons for "Web Dynpro Component" (selected), "Web Dynpro Component Interface", and "Web Dynpro Model Component".
- Window Name:** Z_TUTORIAL2

At the bottom, there are two buttons: a green checkmark and a red X.


Now the WDA Object has been created.



Step 2 – Define Used Components


Define the used web dynpro components for the application. Double click on the web dynpro component. Add the component under the "Used Components" table. Enter SELECT_OPTIONS as the Component Use, and WDR_SELECT_OPTIONS as the Component.

Web Dynpro Component	Z_TUTORIAL2		Inactive/revised
Description	Tutorial 2		
Assistance Class			
Created By	BCUSER	Created On	02/08/2006
Last Changed By	BCUSER	Changed On	02/08/2006
Package	\$TMP	AccessibilityChecks Active	<input checked="" type="checkbox"/>

Used Components		Implemented interfaces
		
Used Web Dynpro Components		
Component Use	Component	Description of Component
SELECT_OPTIONS	WDR_SELECT_OPTIONS	

Step 3 – Creating a View

Create a view by right clicking on the object name and choose Create->View. Give the view name as "MAIN" and a description. Click on the properties tab of the view. In the "Used Controllers/Components" tab, click the create icon. Add the two components for Select_Options.




The screenshot shows the 'Properties' tab of a view named 'MAIN'. The view is currently 'Inactive'. The description is 'Main View'. The lifetime is set to 'framework cont...'. The view was created by 'BCUSER' on '19.12.2005' and last changed by 'BCUSER' on '20.12.2005'. Below the metadata, there is a section titled 'Used Controllers/Components' which contains a table with three columns: 'Component Use', 'Component', and 'Controller'.

Component Use	Component	Controller
	Z_TUTORIAL2	COMPONENTCONTROLLER
SELECT_OPTIONS	WDR_SELECT_OPTIONS	
SELECT_OPTIONS	WDR_SELECT_OPTIONS	INTERFACECONTROLLER

WDA Tutorial II: Using Select Options in a WDA Application

Click on the Context tab of the view. Create a node called "FLIGHTS". Set the Cardinality as 0..N. Next create the attributes by right clicking on the node and choosing Create Using the Wizard->Attributes from Components of Structure. Enter SFLIGHT as the structure to be copied. Select all fields except MANDT and hit enter. The Context should now look like this.



WDA Tutorial II: Using Select Options in a WDA Application

Click on the Layout tab of the view. Add UI element of type ViewContainerUIElement, give the name as "View_Container". Add a button to the view, call it "BUTTON1", and create an Action called "CONTINUE" in the properties box of the Button.

Property	Value	Binding
Properties (Button)		
ID	BUTTON1	
design	standard	
enabled	<input checked="" type="checkbox"/>	
explanation		
imageFirst	<input checked="" type="checkbox"/>	
imageSource		
text	Continue	
textDirection	inherit	
tooltip		
visible	visible	
width		
Events		
onAction		
Layout Data (FlowData)		
cellDesign	padless	
vGutter	none	

Create Action

Component	Z_TUTORIAL2
View	MAIN
Action	Continue
Description	Continue


Select an outbound plug or enter an outbound plug for leaving the view by selecting the pushbutton

Outbound Plug


☒ ☐ ☐

WDA Tutorial II: Using Select Options in a WDA Application

Finally add a table to the layout, name it as "FLIGHTTABLE", define the datasource as the FLIGHTS node from the view controller. Bind the table to the view context by right clicking on the FLIGHTTABLE in the ROOTUIELEMENTCONTAINER. Select all, set the standard cell editor as TextView.





The layout of the view should now look like this.



Step 4 – Defining Attributes

Click on the Attributes tab of the view. Add the following attributes as seen here.



View			MAIN	Inactive
Properties Layout Inbound Plugs Outbound Plugs Cor				
				
Attribute	RefTo	Associated Type		
WD_CONTEXT	<input checked="" type="checkbox"/>	IF_WD_CONTEXT_NODE		
WD_THIS	<input checked="" type="checkbox"/>	IF_MAIN		
WD_COMP_CONTROLLER	<input checked="" type="checkbox"/>	IG_COMPONENTCONTROLLER		
M_HANDLER	<input checked="" type="checkbox"/>	IF_WD_SELECT_OPTIONS		
M_WD_SELECT_OPTIONS	<input checked="" type="checkbox"/>	IWCWDR_SELECT_OPTIONS		
	<input type="checkbox"/>			

Step 5 – Method Implementation

Click on the Methods tab of the view. Double click on the WDDOINIT method. Add the code as you see here:.

```
method WDDOINIT .

data:
    lt_range_table      type ref to data,
    rt_range_table      type ref to data,
    read_only           type abap_bool,
    typename            type string.

data:
    lr_componentcontroller type ref to ig_componentcontroller,
    l_ref_cmp_usage type ref to if_wd_component_usage.

* create the used component
l_ref_cmp_usage = wd_this->wd_cpuse_select_options( ).
if l_ref_cmp_usage->has_active_component( ) is initial.
    l_ref_cmp_usage->create_component( ).
endif.

* get a pointer to the interface controller of the select options
*component
wd_this->m_wd_select_options =
    wd_this->wd_cpifc_select_options( ).

* init the select screen
wd_this->m_handler =
    wd_this->m_wd_select_options->init_selection_screen( ).

* create a range table that consists of this new data element
lt_range_table =
    wd_this->m_handler->create_range_table(
        i_typename = 'S_CARR_ID' ).
```

```
* add a new field to the selection
wd_this->m_handler->add_selection_field(
    i_id = 'S_CARR_ID'
    it_result = lt_range_table
    i_read_only = read_only ).

* create a range table that consists of this new data element
lt_range_table =
    wd_this->m_handler->create_range_table(
        i_typename = 'S_CONN_ID' ).

* add a new field to the selection
wd_this->m_handler->add_selection_field(
    i_id = 'S_CONN_ID'
    it_result = lt_range_table
    i_read_only = read_only ).

endmethod.
```

Click on the Methods List button. Double click on the ONACTIONCONTINUE method. Add the code as you see here.

```
method onactioncontinue .

    data: node_flights    type ref to if_wd_context_node.
    data: rt_carrid type ref to data.
    data: rt_connid type ref to data.
    data: isflight type table of sflight.
    data: wsflight type sflight.
    field-symbols: <fs_carrid> type table,
                  <fs_connid> type table.

    * Retrieve the data from the select option
    rt_carrid = wd_this->m_handler->get_range_table_of_sel_field(
                        i_id = 'S_CARR_ID' ).

    * Assign it to a field symbol
    assign rt_carrid->* to <fs_carrid>.

    * Retrieve the data from the select option
    rt_connid = wd_this->m_handler->get_range_table_of_sel_field(
                        i_id = 'S_CONN_ID' ).

    * Assign it to a field symbol
    assign rt_connid->* to <fs_connid>.


    * Retrieve that data from the database. Normally it is suggested to
    * encapsulate the data retrieval in a separate class.
    * For simplicity, the SELECT statement has been implemented here.
    clear isflight. refresh isflight.
    select * into corresponding fields of table isflight from sflight
           where carrid in <fs_carrid>
           and connid in <fs_connid>.

    * Bind the data to the context
    node_flights = wd_context->get_child_node( name = `FLIGHTS` ).
    node_flights->bind_elements( isflight ).
```

endmethod.

Step 6 – Embed View into Window

Double-click on the window under the Windows folder in the object tree to the left. Drag and drop the MAIN view to the Window.




Expand the window, right click on VIEW_CONTAINER and select Embed View. In the dialog hit F4 on the "View to be embedded" Select the WND_SELECTION_SCREEN from the help.

WDA Tutorial II: Using Select Options in a WDA Application

Component	Z_TUTORIAL2
Window	Z_TUTORIAL2
Container / Area	VIEW_CONTAINER_VCA_1
View to Be Embedded	WND_SELECTION_SCREEN
Component of View	WDR_SELECT_OPTIONS
Component Use	SELECT_OPTIONS

The Window should now look like this.



Step 7 – Create the Application and Test

Finally, create the application and test it. Make sure that you save and activate the web dynpro component. When the browser opens, enter some values in your select-options and click continue. You will see the data in the table below. If you have no data in your Sneak Preview system, run program SAPBC_DATA_GENERATOR to generate the data for you.

WDA Tutorial II: Using Select Options in a WDA Application

Airline	Flight Number	Date	Airfare	Airline Currency	Plane Type	Max. capacity econ.
AA	0017	01.06.2005	422,94	USD	747-400	385
AA	0017	10.08.2005	422,94	USD	747-400	385
AA	0017	19.10.2005	422,94	USD	747-400	385
AA	0017	28.12.2005	422,94	USD	747-400	385
AA	0017	08.03.2006	422,94	USD	747-400	385

Author Bio



Rich Heilman is an ABAP/J2EE Software Engineer/Analyst for Yorktowne Cabinetry, Inc. based in Red Lion, Pennsylvania, USA. He has a total of nine years experience in the IT industry. He has spent the past five years studying ABAP and Java.

Disclaimer & Liability Notice

This document may discuss sample coding or other information that does not include SAP official interfaces and therefore is not supported by SAP. Changes made based on this information are not supported and can be overwritten during an upgrade.

SAP will not be held liable for any damages caused by using or misusing the information, code or methods suggested in this document, and anyone using these methods does so at his/her own risk.

SAP offers no guarantees and assumes no responsibility or liability of any type with respect to the content of this technical article or code sample, including any liability resulting from incompatibility between the content within this document and the materials and services offered by SAP. You agree that you will not hold, or seek to hold, SAP responsible or liable with respect to the content of this document.